Monte Carlo Integration

Clear all symbols from previous evaluations to avoid problems

In[ ]:=
```
Clear["Global`*"]
```

Straightforward implementation without the weight function

Simple Monte Carlo method approximates a multivariable integral by

$$I = \int_{\Omega} f(x_1, \ldots, x_d)\, d^d x = \frac{|\Omega|}{N} \sum_{i=1}^{N} f\left(x_1^{(i)}, \ldots, x_d^{(i)}\right)$$

where $|\Omega|$ denotes the volume of the region Ω.
The following function approximates a *dim*-dimensional integral of a function *f* over the hypercube with boundaries given by *dim*-dimensional vectors *a* and *b.* It returns as an output an array with 3 rows (the number of points, approximation of the integral, error estimation). The number of columns is *exp,* while the number of points for each approximation is given by $N = base^{exp}$. The error estimation is obtain as

$$\text{probable error} = 0.67 \frac{|\Omega|}{\sqrt{N}} \left(\frac{1}{N} \sum_{i=1}^{N} f\left(x_1^{(i)}, \ldots, x_d^{(i)}\right)^2 - \left(\frac{1}{N} \sum_{i=1}^{N} f\left(x_1^{(i)}, \ldots, x_d^{(i)}\right) \right)^2 \right)^{1/2}$$

In[ ]:=
```
mySimpleMCIntND[dim_, f_, a_, b_, base_, exp_] :=
  Module[{i, ie, rndPt, volume, I, II, ip, n, out},
   out = ConstantArray[0, {3, exp}];
   volume = Times @@ (b - a);
   n = base^exp;  (* the largest number of points *)
   I = 0; II = 0;
   ie = 1;  (* exponent, only results for base^ie points are returned *)
   Do[
    rndPt = a + (b - a) RandomReal[{0.0, 1.0}, dim];
    I = I + f[rndPt];  (* sum of function values *)
    II = II + f[rndPt]^2; (* sum of squares of function values *)
    If[Mod[i, base^ie] == 0,
     out[[All, ie]] = {i, volume * I / i, 0.67 * volume * Sqrt[(II / i - (I / i)^2) / i]};
     ie++;
    ],
    {i, 1, n}
   ];
   Return[out];
  ];
```

The volume of the unit ball in *dim* dimensions:

In[ ]:=
```
(* the function to integrate, Norm[x] returs the length of vector x *)
f1[x_] := If[Norm[x] ≤ 1.0, 1.0, 0.0];
dim = 3; (* dimension *)
a = ConstantArray[-1, {dim}];
b = -a; (* boundaries, hypercube arround the origin *)
int = mySimpleMCIntND[dim, f1, a, b, 4, 10]; (* the largest number of points is 4^10 *)
MatrixForm[Transpose[int]]
Print["Exact: hypercube volume = ", N[2^dim],
  ", unit ball volume = ", N[π^(dim/2) / Gamma[1 + dim / 2]]];
```

Out[ ]//MatrixForm=

4	2.	1.160474041
16	4.	0.67
64	3.75	0.3343450629
256	4.25	0.1671725315
1024	4.4609375	0.08319208518
4096	4.197265625	0.04182404674
16 384	4.208984375	0.0209089044
65 536	4.186523438	0.01045736199
262 144	4.193328857	0.005228257668
1 048 576	4.192085266	0.00261416808

Exact: hypercube volume = 8., unit ball volume = 4.188790205

In[•]:=
```
ListLogLogPlot[{Transpose[{int[[1, All]], int[[3, All]]}],
  Transpose[{int[[1, All]], Abs[int[[2, All]] - π^(dim/2) / Gamma[1 + dim / 2]]}]},
 PlotLegends → {"Estimated Error", "Actual Error"}]
```

Out[•]=



Integration of the function 1/(1+x^2)

In[•]:=
```
f[x_] := 1 / (1 + Norm[x] ^2);
dim = 1;
a = ConstantArray[0, {dim}]; b = ConstantArray[1, {dim}];
int = mySimpleMCIntND[dim, f, a, b, 4, 10];
MatrixForm[Transpose[int]]
Print["Exact: ", N[π / 4, 10], ", Error: ", N[Abs[int[[2, 10]] - π / 4], 10]];
```

Out[•]//MatrixForm=

4	0.7345903913	0.05961576813
16	0.8036838258	0.02772840336
64	0.7745706863	0.01409479475
256	0.7775891869	0.007006347236
1024	0.7862734826	0.003432877954
4096	0.7846087347	0.001693250893
16 384	0.7858879031	0.0008417753686
65 536	0.7853635203	0.0004212645321
262 144	0.7855265544	0.0002105751913
1 048 576	0.7854871558	0.0001053007117

```
Exact: 0.7853981634, Error: 0.00008899240117
```

```
In[•]:= ListLogLogPlot[
  {Transpose[{int[[1, All]], int[[3, All]]}], Transpose[{int[[1, All]], Abs[int[[2, All]] - π / 4]}]},
  PlotLegends → {"Estimated Error", "Actual Error"}]
```

Out[•]=



Integration of the function 1/(1+x^2) with the linear function as a weight function

```
In[•]:= f[x_] := 1 / (1 + Norm[x] ^2);
w[x_] := (4 - 2 Norm[x]) / 3;
(* normalized linear function used as a weight function *)
fx[y_] := 2 - Sqrt[4 - 3 Norm[y]]; (* inverse of the CDF of the density w(x) *)
f1[y_] := f[fx[y]] / w[fx[y]]; (* actual function to integrate *)
a = {0}; b = {1};
dim = 1;
int = mySimpleMCIntND[dim, f1, a, b, 4, 10];
{Plot[{f[{x}], w[{x}], f[{x}] / w[{x}]}, {x, a[[1]], b[[1]]}, PlotRange → {0, 1.5},
  PlotLegends → {"f(x)", "w(x)", "f(x)/w(x)"}], MatrixForm[Transpose[int]]}
Print["Exact: ", N[π / 4, 10], ", Error: ", N[Abs[int[[2, 10]] - π / 4], 10]];
```

Out[•]= {



,

4	0.7843935566	0.006918255879
16	0.7887456154	0.003135332198
64	0.781103419	0.001714895418
256	0.7834684374	0.0008191610945
1024	0.7845800408	0.000420341672
4096	0.7851945922	0.0002106343627
16 384	0.7854179602	0.0001049805224
65 536	0.7854102561	0.0000523459745
262 144	0.7853944016	0.00002616070817
1 048 576	0.785406781	0.00001309144601

}

Exact: 0.7853981634, Error: $8.617611829 \times 10^{-6}$

In[ ]:=
```
ListLogLogPlot[
 {Transpose[{int⟦1, All⟧, int⟦3, All⟧}], Transpose[{int⟦1, All⟧, Abs[int⟦2, All⟧ - π / 4]}]},
 PlotLegends → {"Estimated Error", "Actual Error"}]
```

Out[ ]=